Taux de variation moyen et interprétation graphique

© Pierre Lantagne
Enseignant retraité du Collège de Maisonneuve

La première version de ce document est parue sous la version Maple 6. La présente actualisation se présente en deux parties. Dans la première partie, le lecteur sera initié à quelques fonctionnalité Maple et à des éléments de syntaxe qui permettrons de bien documenter les graphiques pour illustrer le concept du taux de variation moyen. La seconde partie se concentre sur les simplifications algébriques du taux de variation moyen portant sur certains types de fonctions algébiques usuelles: fonction polynomiale, fonction rationnelle et la fonction racine carré.

Bonne lecture à tous !

* Ce document Maple est exécutable avec la version 2020.1

Initialisation

```
> restart;
> with(plots,display,setoptions):
  setoptions(size=[300,300],axesfont=[times,roman,8],color=navy):
```

Première partie

Taux de variation moyen



Le taux de variation moyen d'une fonction f sur un intervalle $[a, b]$, noté $TVM_{[a,\,b]}$ est définie comme suit:

$$TVM_{[a,\,b]} = \frac{f(b) - f(a)}{b - a}$$

Graphiquement, le taux de variation moyen d'une fonction f sur un intervalle $[a, b]$ correspond à la pente de la sécante passant par les points $(a, f(a))$ et $(b, f(b))$

À l'aide de la notation indicielle, transposons en Maple le calcul d'un TVM sur un intervalle $[a, b]$ pour une fonction f quelconque. Donnons le nom P au point $(a, f(a))$ et le nom Q au point $(b, f(b))$.

```
> P:=[a,f(a)];
```

```
Q:=[b,f(b)];
TVM[[a,b]]=(Q[2]-P[2]) / (Q[1]-P[1]);
```

$$P := [a, f(a)]$$

$$Q := [b, f(b)]$$

$$TVM_{[a,\, b]} = \frac{f(b) - f(a)}{b - a} \tag{2.1.1}$$

Le membre de gauche de la requête précédente est utile seulement pour une question de mise en forme: la notation indicielle est ici utilisée qu'à la seule fin de transposer, en Maple, le calcul du TVM comme on le fait habituellement en classe.

Soit la fonction f définie par $f(x) = \frac{1}{x+2}$. Exprimons le calcul du $TVM_{\left[-\frac{1}{2},\, 3\right]}$. de la fonction f

```
> f:=x->1/(x+2);
  P:=[-1/2,f(-1/2)];
  Q:=[3,f(3)];
  TVM[[-1/2,3]]=(Q[2]-P[2]) / (Q[1]-P[1]);
```

$$f := x \mapsto \frac{1}{x+2}$$

$$P := \left[-\frac{1}{2}, \frac{2}{3}\right]$$

$$Q := \left[3, \frac{1}{5}\right]$$

$$TVM_{\left[-\frac{1}{2},\, 3\right]} = -\frac{2}{15} \tag{2.1.2}$$

Dans le cas où il y a plusieurs calculs successifs de TVM à effectuer, il est plus commode de créer une fonction de calcul d'un TVM. La fonction que nous allons maintenant créée sera une fonction à deux variables dont les variables seront les points P et Q.

```
> TVM:=(P,Q)->( Q[2]-P[2] ) / (Q[1]-P[1]);
```

$$TVM := (P, Q) \mapsto \frac{Q_2 - P_2}{Q_1 - P_1} \tag{2.1.3}$$

Obtenons à nouveau le $TVM_{\left[-\frac{1}{2},\, 3\right]}$. de la fonction f. Puisque les points $P\left(-\frac{1}{2}, f\left(-\frac{1}{2}\right)\right)$ et $Q(3, f(3))$ ont déjà été créés, il vous suffit d'appliquer la nouvelle fonction à deux variables TVM avec les arguments P et Q.

```
> TVM(P,Q);
```

$$-\frac{2}{15} \tag{2.1.4}$$

Graphiquement, la valeur $\frac{2}{15}$, soit le $TVM_{\left[-\frac{1}{2},\, 3\right]}$ correspond à la pente de la sécante passant par les points $P\left(-\frac{1}{2}, f\left(-\frac{1}{2}\right)\right)$ et $Q(3, f(3))$

Nous allons maintenant illustrer cette inteprétaion. Traçons la fonction f pour $x \in [-4, 4]$. Il faut évidemment contrôler l'affichage de l'axe des y car sur cet intervalle, le dénominateur de la fonction f s'annule. Contrôlons donc l'affichage des axes avec l'option `view = [-4..4,-4..4]`

```
> Courbe:=plot([x,f(x),x=-4..4],discont=true):
  display(Courbe,view=[-4..4,-4..4]);
```



Ensuite, superposons la représentation des points P et Q par des petits cercles et reliez-les par un segment.

```
> Segment:=plot([P,Q],style=line,color=khaki):
  Triangle:=plot([P,[P[1],Q[2]],Q],style=line,linestyle=2,color=
  black):
  Points:=plot([P,Q],style=point,symbol=circle,color=navy):
  display({Courbe,Segment,Triangle,Points},view=[-4..4,-4..4]);
```



Modifions l'option view afin de mieux mettre en évidence le segment.

```
> display({Courbe,Segment,Triangle,Points},view=[-1..4,0..1]);
```



Il est possible de documenter un graphique avec du texte, c'est-à-dire avec des objets Maple de type `string`. La macro-commande `textplot` de l'extension `plots` a été conçue à cet effet. La syntaxe de base est la suivante:

`textplot([x,y,texte],alignement)`

Le premier argument est une `liste` de trois éléments. Le premier et le deuxième élément précise les coordonnées du point de départ de l'écriture et le troisième élément étant le texte à écrire. Par défaut, l'alignement est un alignement centré, sinon, il est possible d'imposer l'un des alignements BELOW, RIGHT, ABOVE, LEFT.

Superposons les noms et coordonnées des points P et Q au tracé précédent.

ATTENTION: Par les assignations précédentes, les lettres P et Q, pointant respectivement sur leurs coordonnées, sont des objets de type `liste` et ne seront donc pas traités par l'évaluateur comme de simples lettres. Il nous faudra donc convertir chacune de ces deux liste de coordonnées en une chaîne de caractères à l'aide de la macro-commande `convert`.

De plus, afin de faire afficher les noms P et Q à côté de leurs coordonnées, il faudra réaliser ce qu'on appelle une concaténation d'expressions à l'aide de la macro-commande `cat`.

Avant d'être en mesure d'utiliser la macro-commande `textplot`, il faut d'abord la rendre disponible pour la durée de cette session en saisissant la requête `with(plots,textplot)`.

```
> with(plots,textplot);
```

$$[textplot] \quad (2.1.5)$$

```
> Texte_P:=textplot([P[1],P[2],cat("P",convert(P,string))],align=
  {BELOW,LEFT}):
  Texte_Q:=textplot([Q[1],Q[2],cat("Q",convert(Q,string))],align=
  {ABOVE,RIGHT}):
  display({Courbe,Segment,Triangle,Points,Texte_P,Texte_Q},view=[-1.
  .4,0..1]);
```



Au lieu de relier les points P et Q par un segment de droite, traçons donc plutôt une véritable droite passant par ces deux points.

Rappelons que l'équation d'une droite de pente m passant par le point $Q\left(x_0, y_0\right)$ est de la forme

$$y = m\left(x - x_0\right) + y_0$$

Le point $P\left(-\frac{1}{2}, \frac{2}{3}\right)$ appartenant à cette sécante (l'autre point, $Q\left(3, \frac{1}{5}\right)$, aurait pu aussi bien fait l'affaire), l'équation de la sécante est

$$y = m_{\text{sec}}\left(x - \left(-\frac{1}{2}\right)\right) + \frac{2}{3}$$

Puisque la pente $m_{\mathrm{sec}} = TVM_{\left[-\frac{1}{2}, 3\right]}$, l'équation de la droite sécante est donc donnée, en syntaxe Maple, par

$$y = TVM(P, Q)\ (x - P[1]) + P[2]$$

```
> Éq_Sécante:=y=TVM(P,Q)*(x-P[1])+P[2];
```

$$Éq\_Sécante := y = -\frac{2x}{15} + \frac{3}{5} \qquad \mathbf{(2.1.6)}$$

Montrons que le point Q aurait pu tout aussi bien fait l'affaire pour obtenir l'équation de cette sécante.

```
> Éq_Sécante:=y=TVM(P,Q)*(x-Q[1])+Q[2];
```

$$Éq\_Sécante := y = -\frac{2x}{15} + \frac{3}{5} \qquad \mathbf{(2.1.7)}$$

Superposons cette dernière droite au tracé au lieu du segment reliant les points P et Q.

```
> Droite:=plot([x,rhs(Éq_Sécante),x=-3..4],color=khaki):
  display({Courbe,Droite,Triangle,Points,Texte_P,Texte_Q},view=[-1.
  .4,0..1]);
```



Dans le contexte des taux de variation moyen, la notation de l'intervalle $[a, b]$ est habituellement reformulée par la notation $[a, a+h]$ avec, évidemment, $h > 0$. Ainsi, la valeur de b, notée $a + h$, mets implicitement en évidence un *accroissement positif h* qui est donné à a pour obtenir la borne supérieure b de l'intervalle $[a, b]$.

On obtient alors la définition suivante du taux de variation moyen à droite.

Définition
Le taux de variation moyen à droite de $(a, f(a))$ noté $TVM_{[a, a+h]}$ est définie par

$$TVM_{[a, a+h]} = \frac{f(a+h) - f(a)}{h}$$

Appliquons donc la fonction TVM que nous avant créée avec cette notation.

Désassignons d'abord le nom f pour transposer en Maple cette définition du $TVM_{[a, a+h]}$.

```
> f:='f':
  P:=[a,f(a)];
  Q:=[a+h,f(a+h)];
  TVM[[a,a+h]]=TVM(P,Q);
```

$$P := [a, f(a)]$$

$$Q := [a+h, f(a+h)]$$

$$TVM_{[a,\, a+h]} = \frac{f(a+h) - f(a)}{h} \tag{2.1.8}$$

Plus généralement, le taux de variation moyen à droite d'une fonction continue f se définit de la façon suivante.

Définition
Le taux de variation moyen à droite de $(x, f(x))$ d'une fonction f , noté $TVM_{[x, x+h]}$ est définie par

$$TVM_{[x, x+h]} = \frac{f(x+h) - f(x)}{h}$$

Remarque: La notation elle-même de $TVM_{[x, x+h]}$ précise qu'il s'agit du taux de variation à droite car la notation de l'intervalle $[x, x+h]$ impose que la valeur h soit positive puisque que l'on puisse avoir $x < x+h$.

Reformulons donc la fonction TVM avec cette notation.

```
> P:=[x,f(x)];
  Q:=[x+h,f(x+h)];
  TVM[[x,x+h]]=TVM(P,Q);
```

$$P := [x, f(x)]$$

$$Q := [x+h, f(x+h)]$$

$$TVM_{[x,\, x+h]} = \frac{f(x+h) - f(x)}{h} \tag{2.1.9}$$

On aurait pu aussi, reformuler l'intervalle $[a, b]$ avec la notation $[b+h, b]$, avec, évidemment, $h < 0$. Ainsi, la valeur de a, notée $b+h$, mets en évidence *accroissement négatif* h qui est donné à b pour obtenir la borne inférieure a de l'intervalle $[a, b]$.

On obtient alors la définition suivante du taux de variation moyen à gauche.

Définition
Le taux de variation moyen à gauche de $(b, f(b))$ fonction f , noté $TVM_{[b+h, b]}$ est définie par

$$TVM_{[b+h, b]} = \frac{f(b) - f(b+h)}{-h}$$
$$= \frac{f(b+h) - f(b)}{h}$$

Et, plus généralement, le taux de variation moyen à gauche d'une fonction continue f se définit de la façon suivante.

Définition
Le taux de variation moyen à gauche de $(x, f(x))$ d'une fonction f , noté $TVM_{[x+h, x]}$ est définie par

$$TVM_{[x+h, x]} = \frac{f(x) - f(x+h)}{-h}$$
$$= \frac{f(x+h) - f(x)}{h}$$

La formule algébrique du taux de variation moyen à gauche et à droite est donc la même:

$$TVM_{[x+h,x]} = \frac{f(x+h)-f(x)}{h} = TVM_{[x,x+h]}$$

En effet, dans les deux cas, il s'agit en fait du même calcul de la pente de la sécante passant par les points $(x, f(x))$ et $(x+h, f(x+h))$

Ce qui distinguera le calcul d'un taux de variation moyen à gauche d'avec le calcul d'un taux de variation à droite, c'est le signe algébrique de l'accroissement *h* donné à *x*.

Seconde partie

Calcul de taux de variation moyen

Avec la macro-command `restart` , initialisons notre environnement à celui par défaut.

```
> restart;
```

Le calcul du taux de variation moyen à gauche ou à droite sera toujours obtenu par même calcul algébrique de la pente de la sécante passant par les points $(x, f(x))$ et $(x+h, f(x+h))$.

Saisissons à nouveau la macro-commande TVM qui donnera le taux de variation moyen.

```
> P:=[x,f(x)];
  Q:=[x+h,f(x+h)];
  TVM:=(P,Q)->( Q[2]-P[2] ) / (Q[1]-P[1]);
  TVM[[x,x+h]]=TVM(P,Q);
```

$$P := [x, f(x)]$$

$$Q := [x+h, f(x+h)]$$

$$TVM := (P, Q) \mapsto \frac{Q_2 - P_2}{Q_1 - P_1}$$

$$TVM_{[x,\, x+h]} = \frac{f(x+h)-f(x)}{h} \tag{3.1.1}$$

Le calcul de TVM(*P*, *Q*) exprimera le taux de variation moyen à la droite de *P* tandis que le calcul de TVM (*Q*, *P*) exprimera le taux de variation à la gauche de *P*.

Montronsque TVM(*Q*, *P*) = TVM(*P*, *Q*).

```
> TVM[[x+h,x]]=TVM(Q,P);
  ``=normal(rhs(%),expanded);
```

$$TVM_{[x+h,\, x]} = -\frac{f(x)-f(x+h)}{h}$$

$$= \frac{f(x+h)-f(x)}{h} \tag{3.1.2}$$

Dans les exemples suivants de calculs d'un TVM, afin d'alléger le texte, il s'agira tout le temps du taux de variation moyen à droite de $(x, f(x))$. Aussi, il nous faudra simplifier algébriquement l'expression du TVM. Notre objectif de simplification sera atteint seulement lorsque le facteur *h* sera éléminé.

Exemple 1

Calculons le $TVM_{[x,\, x+h}$ avec la fonction f définie par $f(x) = 3x^2-5$.

```
> f:=x->3*x^2-5;
  TVM[[x,x+h]]=TVM(P,Q);
```

$$f := x \mapsto 3 \cdot x^2 - 5$$

$$TVM_{[x,\, x+h]} = \frac{3(x+h)^2 - 3x^2}{h} \qquad \textbf{(3.1.1.1)}$$

Reste plus qu'à simplifier le $TVM_{[x,\, x+h]}$ afin d'éliminer le facteur h au dénominateur. Appliquons une simplification sur demande avec la macro-commande `normal`.

```
> normal((3.1.1.1));
```

$$TVM_{[x,\, x+h]} = 3h + 6x \qquad \textbf{(3.1.1.2)}$$

Exemple 2

Soit à calculer le $TVM_{[x,\, x+h]}$ pour la fonction f définie par $f(x) = 2x^3 - 5$.

```
> f:=x->2*x^3-5;
  TVM[[x,x+h]]=TVM(P,Q);
```

$$f := x \mapsto 2 \cdot x^3 - 5$$

$$TVM_{[x,\, x+h]} = \frac{2(x+h)^3 - 2x^3}{h} \qquad \textbf{(3.1.2.1)}$$

Reste à simplifier

```
> normal((3.1.2.1));
```

$$TVM_{[x,\, x+h]} = 2h^2 + 6xh + 6x^2 \qquad \textbf{(3.1.2.2)}$$

Exemple 3

Soit à calculer le $TVM_{[x,\, x+h]}$ pour la fonction f définie par $f(x) = \dfrac{x^2+1}{2x-5}$.

```
> f:=x->(x^2+1)/(2*x-5);
  TVM[[x,x+h]]=TVM(P,Q);
```

$$f := x \mapsto \frac{x^2+1}{2 \cdot x - 5}$$

$$TVM_{[x,\, x+h]} = \frac{\dfrac{(x+h)^2+1}{2x+2h-5} - \dfrac{x^2+1}{2x-5}}{h} \qquad \textbf{(3.1.3.1)}$$

```
> normal((3.1.3.1));
```

$$TVM_{[x,\, x+h]} = \frac{2xh + 2x^2 - 5h - 10x - 2}{(2x+2h-5)(2x-5)} \qquad \textbf{(3.1.3.2)}$$

Exemple 4

Le prochain exemple va exiger un plus grand déploiement d'efforts pour simplifier correctement le TVM (afin d'éliminer le facteur h).

Soit à calculer le $TVM_{[x,\, x+h]}$ pour la fonction f définie par $f(x) = \sqrt{2x+3}$.

```
> f:=x->sqrt(2*x+3);
  TVM[[x,x+h]]=TVM(P,Q);
```

$$f := x \mapsto \sqrt{2 \cdot x + 3}$$

$$TVM_{[x,\, x+h]} = \frac{\sqrt{2x+2h+3} - \sqrt{2x+3}}{h} \qquad (3.1.4.1)$$

```
>  radnormal((3.1.4.1),rationalized);
```

$$TVM_{[x,\, x+h]} = -\frac{-\sqrt{2x+2h+3} + \sqrt{2x+3}}{h} \qquad (3.1.4.2)$$

C'est bien pensé mais la macro-commande `radnormal` n'a pas été efficace dans ce cas-ci. En effet, `radnormal` n'est pas la macro-commande appropriée pour obtenir la simplification désirée car cette macro-commande opère seulement la simplification de nombres exprimés avec des radicaux.

La macro-commande `rationalize` de la bibliothèque principale permet de *rationaliser le dénominateur* d'une expression en générale, pas seulement celles avec de nombres exprimés avec des radicaux.

Soyons astucieux. Puisque la simplification du TVM exige la rationalisation du numérateur, nous allons, tout simplement, rationaliser d'abord l'inverse multiplicatif du TVM obtenu puis d'inverser cette rationalisation. Développons cette démarche doucement.

```
>  1/TVM(P,Q);
```

$$\frac{h}{\sqrt{2x+2h+3} - \sqrt{2x+3}} \qquad (3.1.4.3)$$

```
> rationalize(1/TVM(P,Q));
```

$$\frac{\sqrt{2x+3}}{2} + \frac{\sqrt{2x+2h+3}}{2} \qquad (3.1.4.4)$$

```
> 1/rationalize(1/TVM(P,Q));
```

$$\frac{1}{\frac{\sqrt{2x+3}}{2} + \frac{\sqrt{2x+2h+3}}{2}} \qquad (3.1.4.5)$$

```
> TVM[[x,x+h]]=normal((3.1.4.5));
```

$$TVM_{[x,\, x+h]} = \frac{2}{\sqrt{2x+3} + \sqrt{2x+2h+3}} \qquad (3.1.4.6)$$

Bingo !